Apparently-To: catholic@american.edu
Newsgroups: bit.listserv.catholic
Date:         Fri, 22 Jan 1993 14:05:40 PST
From:         "John R. Covert 22-Jan-1993 1656" <covert@COVERT.ENET.DEC.COM>
Subject:      RE: eastern rite catholics

>Do Eastern Rite Catholics use the Gregorian calendar (and its date of
>Easter) or the Julian calendar?

I'll leave the full answer to this to someone else; see the note for
Maronite Catholics in the next paragraph.

The discrepancy in the date of Easter between the Eastern and Western
Church has nothing to do with the Gregorian vs. Julian calendar; it
has to do with how the 14th day of the lunar month is calculated, with
eastern tradition starting the day in the evening and western tradition
starting the day at midnight.  Interestingly, although the Maronite day
starts in the evening (keeping semitic tradition), Maronites use the
Western calculation for the date of Easter.

The date of Easter has been an issue throughout the ages.  Although the
calculation of the date of Easter which the Western Church uses was agreed
upon at the council of Nicaea, much of the Church didn't go along with the
change.  For example, the Church in England used the eastern calculation for
Easter until it was latinized in the sixth century.

To this reply I have appended a VMS DCL procedure which implements the
Western Church calculations just provided for both the Gregorian and
Julian calendar.  If invoked for dates prior to 600, it issues a warning
about the East/West controversy; for dates prior to 1583 it uses the
Julian calculation; if invoked for dates from 1583 to 1923 it provides
both dates with some info about when and where the Gregorian calendar
was adopted; after 1923 it only provides the Gregorian date.

Does anyone have similar formulas for the Eastern Church?

/john

$ ! EASTER.COM
$ !
$ ! Finds the date of Easter (Western Church) -- Julian or Gregoran calendar
$ !
$ if p1 .eqs. "" .or. p1 .lt. 1
$ then
$   write sys$output "Usage is @EASTER year"
$   exit
$ endif
$ year=p1
$ if year .le. 1582 then goto julian
$!Easter calculation for Gregorian calendar
$!Divide        by      Quotient        Remainder
$!the year      19         -                a
$!the year      100        b                c
$!    b         4          d                e
$!   b+8        25         f                -
$!  b-f+1       3          g                -
$!19a+b-d-g+15  30         -                h
$!    c         4          i                k
$!32+2e+2i-h-k  7          -                L
$!a+11h+22L     451        m                -
$!h+L-7m+114    31         n                p
$!
$!n = number of month (3 = March; 4 = April)
$!p+1 = day of the month of Easter Sunday
$!
$ year=0'p1'
$ a=year-year/19*19
$ b=year/100
$ c=year-b*100
$ d=b/4
$ e=b-d*4
$ f=(b+8)/25
$ g=(b-f+1)/3
$ g2=19*a+b-d-g+15
$ h=g2-g2/30*30
$ i=c/4
$ k=c-i*4
$ k2=32+2*e+2*i-h-k
$ L=k2-k2/7*7
$ m=(a+11*h+22*L)/451
$ m2=h+L+7*m+114
$ n=m2/31
$ p=m2-n*31
$ write sys$output p+1,f$element(n-3,"/"," March, / April, "),year
$ if year .gt. 1923 then exit
$ type sys$input

Prior to 1923, the Julian calendar was still in use in some countries.
The date above is according to the Gregorian calendar, which was first
adopted by Spain and her colonies (including Florida) the day following
Oct 4th, 1582, which was reckoned as Oct 15th.  Gradually, other nations
adopted the Gregorian calendar: the Catholic German states in 1583; the
Protestant German states in 1699; England and her colonies in September
1752; Sweden in 1753; Alaska upon purchase in 1867; Japan in 1873; China
in 1912; the Soviet Union in 1918; and Greece in 1923.  In Switzerland,
both calendars were in use from 1583 until the country was fully converted
in 1812.

The date below is the date for Easter according to the older Julian
calendar.

$julian:
$ if year .lt. 600
$ then type sys$input

The exact year of the first Easter is unknown, but is believed to have
been between 28 and 33 A.D.  The calculation used below, that of the Western
Church, decided at the council of Nicaea in 325, is based on the day starting
at midnight, whereas in Eastern tradition, the day starts the evening before.

England did not adopt the Western date for Easter until some time in the sixth
century, so this calculation is not correct until the Church in England was
Latinized.

$ endif
$!
$!Julian calendar calculation
$!
$!Divide        by      Quotient        Remainder
$!the year      4          -                a
$!the year      7          -                b
$!the year      19         -                c
$!19c+15        30         -                d
$!2a+4b-d+34    7          -                e
$!d+e+114       31         f                g
$!
$!f = number of the month (3 = March; 4 = April)
$!g+1 = day of the month of Easter Sunday
$!
$ a=year-year/4*4
$ b=year-year/7*7
$ c=year-year/19*19
$ c2=19*c+15
$ d=c2-c2/30*30
$ d2=2*a+4*b-d+34
$ e=d2-d2/7*7
$ e2=d+e+114
$ f=e2/31
$ g=e2-f*31
$ write sys$output g+1,f$element(f-3,"/"," March, / April, "),year
Comments: Gated by NETNEWS@AUVM.AMERICAN.EDU